/*******************************************************************************
 * Copyright (c) 2000, 2004 IBM Corporation and others.
 * All rights reserved. This program and the accompanying materials
 * are made available under the terms of the Common Public License v1.0
 * which accompanies this distribution, and is available at
 * http://www.eclipse.org/legal/cpl-v10.html
 * 
 * Contributors:
 *     IBM Corporation - initial API and implementation
 *******************************************************************************/
module dwt.widgets.group;


private import dwt.dwt;

private import dwt.graphics.rectangle;
private import dwt.internal.win32.os;
private import dwt.util.javatypes;
private import dwt.util.util;
public import dwt.widgets.composite;
private import dwt.widgets.shell;


/**
 * Instances of this class provide an etched border
 * with an optional title.
 * <p>
 * Shadow styles are hints and may not be honoured
 * by the platform.  To create a group with the
 * default shadow style for the platform, do not
 * specify a shadow style.
 * <dl>
 * <dt><b>Styles:</b></dt>
 * <dd>SHADOW_ETCHED_IN, SHADOW_ETCHED_OUT, SHADOW_IN, SHADOW_OUT, SHADOW_NONE</dd>
 * <dt><b>Events:</b></dt>
 * <dd>(none)</dd>
 * </dl>
 * <p>
 * Note: Only one of the above styles may be specified.
 * </p><p>
 * IMPORTANT: This class is <em>not</em> intended to be subclassed.
 * </p>
 */

public class Group : Composite {
	static int CLIENT_INSET = 3;
	static WNDPROC_I GroupProc;

	version(WinCE){
		static const char[] GroupClass = "BUTTON"; 
	}else{
	 	static const char[] GroupClass = "DWT_GROUP";
	}
	private static boolean staticCtor_done = false;
	static void staticCtor() {
		if(staticCtor_done) return;
		staticCtor_done = true;

		/*
		* Feature in Windows.  The group box window class
		* uses the CS_HREDRAW and CS_VREDRAW style bits to
		* force a full redraw of the control and all children
		* when resized.  This causes flashing.  The fix is to
		* register a new window class without these bits and
		* implement special code that damages only the exposed
		* area.
		* 
		* Feature in WinCE.  On certain devices, defining
		* a new window class which looks like BUTTON causes
		* CreateWindowEx() to crash.  The workaround is to use
		* the class Button directly.
		*/
		WNDCLASS_T lpWndClass;
		version(WinCE){
			OS.GetClassInfo (null, Converter.StrToTCHARz(GroupClass), &lpWndClass);
			GroupProc = lpWndClass.lpfnWndProc;
		} else {
			TCHAR[] WC_BUTTON = "BUTTON\0";
			OS.GetClassInfo (null, WC_BUTTON.ptr, &lpWndClass);
			GroupProc = lpWndClass.lpfnWndProc;
			HINSTANCE hInstance = OS.GetModuleHandle (null);
			if (!OS.GetClassInfo (hInstance, Converter.StrToTCHARz(GroupClass), &lpWndClass)) {
				lpWndClass.hInstance = hInstance;
				lpWndClass.style &= ~(OS.CS_HREDRAW | OS.CS_VREDRAW);
				lpWndClass.lpszClassName = Converter.StrToTCHARz(GroupClass);
				int result = OS.RegisterClass (&lpWndClass);
			}
		}
	}

/**
 * Constructs a new instance of this class given its parent
 * and a style value describing its behavior and appearance.
 * <p>
 * The style value is either one of the style constants defined in
 * class <code>DWT</code> which is applicable to instances of this
 * class, or must be built by <em>bitwise OR</em>'ing together 
 * (that is, using the <code>int</code> "|" operator) two or more
 * of those <code>DWT</code> style constants. The class description
 * lists the style constants that are applicable to the class.
 * Style bits are also inherited from superclasses.
 * </p>
 *
 * @param parent a composite control which will be the parent of the new instance (cannot be null)
 * @param style the style of control to construct
 *
 * @exception IllegalArgumentException <ul>
 *    <li>ERROR_NULL_ARGUMENT - if the parent is null</li>
 * </ul>
 * @exception SWTException <ul>
 *    <li>ERROR_THREAD_INVALID_ACCESS - if not called from the thread that created the parent</li>
 *    <li>ERROR_INVALID_SUBCLASS - if this class is not an allowed subclass</li>
 * </ul>
 *
 * @see DWT#SHADOW_ETCHED_IN
 * @see DWT#SHADOW_ETCHED_OUT
 * @see DWT#SHADOW_IN
 * @see DWT#SHADOW_OUT
 * @see DWT#SHADOW_NONE
 * @see Widget#checkSubclass
 * @see Widget#getStyle
 */
public this (Composite parent, int style) {
	staticCtor();
	super (parent, checkStyle (style));
}

LRESULT callWindowProc (int msg, int wParam, int lParam) {
	if (handle is null) return 0;
	/*
	* Feature in Windows.  When the user clicks on the group
	* box label, the group box takes focus.  This is unwanted.
	* The fix is to avoid calling the group box window proc.
	*/
	switch (msg) {
		case OS.WM_LBUTTONDOWN:
		case OS.WM_LBUTTONDBLCLK: 
			return OS.DefWindowProc (handle, msg, wParam, lParam);
		default : break;
	}
	return OS.CallWindowProc (GroupProc, handle, msg, wParam, lParam);
}

static int checkStyle (int style) {
	style |= DWT.NO_FOCUS;
	/*
	* Even though it is legal to create this widget
	* with scroll bars, they serve no useful purpose
	* because they do not automatically scroll the
	* widget's client area.  The fix is to clear
	* the DWT style.
	*/
	return style & ~(DWT.H_SCROLL | DWT.V_SCROLL);
}

protected void checkSubclass () {
	if (!isValidSubclass ()) error(__FILE__, __LINE__, DWT.ERROR_INVALID_SUBCLASS);
}

public Rectangle computeTrim (int x, int y, int width, int height) {
	checkWidget ();
	Rectangle trim = super.computeTrim (x, y, width, height);
	HFONT newFont, oldFont;
	HDC hDC = OS.GetDC (handle);
	newFont = cast(HFONT)OS.SendMessage (handle, OS.WM_GETFONT, 0, 0);
	if (newFont !is null) oldFont = OS.SelectObject (hDC, newFont);
	TEXTMETRIC* tm = new TEXTMETRIC();
	OS.GetTextMetrics (hDC, tm);
	int textWidth = 0;
	int _length = OS.GetWindowTextLength (handle);
	if (_length != 0) {
		/*
		* If the group has text, and the text is wider than the
		* client area, pad the width so the text is not clipped.
		*/
		TCHAR[] buffer1 = new TCHAR[_length + 1];
		OS.GetWindowText (handle, buffer1.ptr, _length + 1);
		RECT rect;
		int flags = OS.DT_CALCRECT | OS.DT_SINGLELINE;
		OS.DrawText (hDC, buffer1.ptr, _length, &rect, flags);
		textWidth = rect.right - rect.left + CLIENT_INSET * 4;
	}
	if (newFont !is null) OS.SelectObject (hDC, oldFont);
	OS.ReleaseDC (handle, hDC);
	trim.x -= CLIENT_INSET;
	trim.y -= tm.tmHeight;
	trim.width = Math.max (trim.width, textWidth) + CLIENT_INSET * 2;
	trim.height += tm.tmHeight + CLIENT_INSET;
	return trim;
}

void createHandle () {
	super.createHandle ();
	state &= ~CANVAS;
}

public Rectangle getClientArea () {
	checkWidget ();
	forceResize ();
	RECT rect;
	OS.GetClientRect (handle, &rect);
	HFONT newFont, oldFont;
	HDC hDC = OS.GetDC (handle);
	newFont = cast(HFONT)OS.SendMessage (handle, OS.WM_GETFONT, 0, 0);
	if (newFont !is null) oldFont = OS.SelectObject (hDC, newFont);
	TEXTMETRIC* tm = new TEXTMETRIC();
	OS.GetTextMetrics (hDC, tm);
	if (newFont !is null) OS.SelectObject (hDC, oldFont);
	OS.ReleaseDC (handle, hDC);
	int x = CLIENT_INSET, y = tm.tmHeight;
	int width = rect.right - CLIENT_INSET * 2;
	int height = rect.bottom - y - CLIENT_INSET;
	return new Rectangle (x, y, width, height);
}

char[] getNameText () {
	return getText ();
}

/**
 * Returns the receiver's text, which is the string that the
 * is used as the <em>title</em>. If the text has not previously
 * been set, returns an empty string.
 *
 * @return the text
 *
 * @exception SWTException <ul>
 *    <li>ERROR_WIDGET_DISPOSED - if the receiver has been disposed</li>
 *    <li>ERROR_THREAD_INVALID_ACCESS - if not called from the thread that created the receiver</li>
 * </ul>
 */
public char[] getText () {
	checkWidget ();
	return OS.GetWindowText (handle);
}

boolean mnemonicHit (wchar key) {
	return setFocus ();
}

boolean mnemonicMatch (wchar key) {
	wchar mnemonic = findMnemonic (getText ());
	if (mnemonic == '\0') return false;
	return Character.toUpperCase (key) == Character.toUpperCase (mnemonic);
}

/**
 * Sets the receiver's text, which is the string that will
 * be displayed as the receiver's <em>title</em>, to the argument,
 * which may not be null. The string may include the mnemonic character.
 * </p>
 * Mnemonics are indicated by an '&amp' that causes the next
 * character to be the mnemonic.  When the user presses a
 * key sequence that matches the mnemonic, focus is assgned
 * to the first child of the group. On most platforms, the
 * mnemonic appears underlined but may be emphasised in a
 * platform specific manner.  The mnemonic indicator character
 *'&amp' can be escaped by doubling it in the string, causing
 * a single '&amp' to be displayed.
 * </p>
 * @param string the new text
 *
 * @exception IllegalArgumentException <ul>
 *    <li>ERROR_NULL_ARGUMENT - if the text is null</li>
 * </ul>
 * @exception SWTException <ul>
 *    <li>ERROR_WIDGET_DISPOSED - if the receiver has been disposed</li>
 *    <li>ERROR_THREAD_INVALID_ACCESS - if not called from the thread that created the receiver</li>
 * </ul>
 */
public void setText (char[] string) {
	checkWidget ();
	if (string !is null) 
		OS.SetWindowText (handle, Converter.StrToTCHARz(string));
}


int widgetStyle () {
	/*
	* Bug in Windows.  When GetDCEx () is called with DCX_INTERSECTUPDATE,
	* the HDC that is returned does not include the current update region.
	* This was confirmed under DEBUG Windows when GetDCEx () complained about
	* invalid flags.  Therefore, it is not easily possible to get an HDC from
	* outside of WM_PAINT that includes the current damage and clips children.
	* Because the receiver has children and draws a frame and label, it is
	* necessary that the receiver always draw clipped, in the current damaged
	* area.  The fix is to force the receiver to be fully clipped by including
	* WS_CLIPCHILDREN and WS_CLIPSIBLINGS in the default style bits.
	*/
	return super.widgetStyle () | OS.BS_GROUPBOX | OS.WS_CLIPCHILDREN | OS.WS_CLIPSIBLINGS;
}

char[] windowClass() {
	return GroupClass;
}

WNDPROC_I windowProc2 () {
	return GroupProc;
}

LDWTRESULT WM_ERASEBKGND_ (int wParam, int lParam) {
	LDWTRESULT result = super.WM_ERASEBKGND_ (wParam, lParam);
	if (result !is null) return result;
	/*
	* Feaure in Windows.  Group boxes do not erase
	* the background before drawing.  The fix is to
	* fill the background.
	*/
	drawBackground (cast(HWND)wParam);
	return LDWTRESULT.ONE;
}

LDWTRESULT WM_NCHITTEST_ (int wParam, int lParam) {
	LDWTRESULT result = super.WM_NCHITTEST_ (wParam, lParam);
	if (result !is null) return result;
	/*
	* Feature in Windows.  The window proc for the group box
	* returns HTTRANSPARENT indicating that mouse messages
	* should not be delivered to the receiver and any children.
	* Normally, group boxes in Windows do not have children and
	* this is the correct behavior for this case.  Because we
	* allow children, answer HTCLIENT to allow mouse messages
	* to be delivered to the children.
	*/
	int code = callWindowProc (OS.WM_NCHITTEST, wParam, lParam);
	if (code == OS.HTTRANSPARENT) code = OS.HTCLIENT;
	return new LDWTRESULT (code);
}

LDWTRESULT WM_MOUSEMOVE_ (int wParam, int lParam) {
	LDWTRESULT result = super.WM_MOUSEMOVE_ (wParam, lParam);
	if (result !is null) return result;
	/*
	* Feature in Windows.  In version 6.00 of COMCTL32.DLL,
	* every time the mouse moves, the group title redraws.
	* This only happens when WM_NCHITTEST returns HTCLIENT.
	* The fix is to avoid calling the group window proc.
	*/
	return LDWTRESULT.ZERO;
}

LDWTRESULT WM_PRINTCLIENT_ (int wParam, int lParam) {
	LDWTRESULT result = super.WM_PRINTCLIENT_ (wParam, lParam);
	if (result !is null) return result;
	/*
	* Feature in Windows.  In version 6.00 of COMCTL32.DLL,
	* when WM_PRINTCLIENT is sent from a child BS_GROUP
	* control to a parent BS_GROUP, the parent BS_GROUP
	* clears the font from the HDC.  Normally, group boxes
	* in Windows do not have children so this behavior is
	* undefined.  When the parent of a BS_GROUP is not a
	* BS_GROUP, there is no problem.  The fix is to save
	* and restore the current font.
	*/
	if (OS.COMCTL32_MAJOR >= 6) {
		int nSavedDC = OS.SaveDC (cast(HDC)wParam);
		int code = callWindowProc (OS.WM_PRINTCLIENT, wParam, lParam);
		OS.RestoreDC (cast(HDC)wParam, nSavedDC);
		return new LDWTRESULT (code);
	}
	return result;
}

LDWTRESULT WM_SIZE_ (int wParam, int lParam) {
	LDWTRESULT result = super.WM_SIZE_ (wParam, lParam);
	version(WinCE) { return result; }
	OS.InvalidateRect (handle, null, true);
	return result;
}

}
